Chapter 8
Calculus

8.1 Series

Let's expand a function in x at the point $x = 0$ up to the fifth order.
In[1] := s = Series[Exp[x], {x, 0, 5}]

$$\text{Out}[1] = 1 + x + \frac{x^2}{2} + \frac{x^3}{6} + \frac{x^4}{24} + \frac{x^5}{120} + O[x]^6$$

How are series represented in *Mathematica*? By the function SeriesData. Its first argument is the expansion variable; the second one—the expansion point; the third one—the list of coefficients; the fourth one—the minimum degree (here 0); the fifth one—the power of $O[x]$; the sixth one is 1 for series with integer degrees (all degrees are divided by it if it's not 1). Thus a series is not a sum (Plus) in spite of its appearance.
In[2] := FullForm[s]
Out[2]//FullForm =

SeriesData[x, 0, List[1, 1, Rational[1, 2], Rational[1, 6], Rational[1, 24], Rational[1, 120]], 0, 6, 1]

Coefficients are extracted by the function SeriesCoefficient.
In[3] := Do[Print[SeriesCoefficient[s, n]], {n, 0, 5}]

1

1

$\frac{1}{2}$

$\frac{1}{6}$

$\frac{1}{24}$

$\frac{1}{120}$

This series begins with degree -1.
In[4] := s = Series[Cot[x], {x, 0, 5}]

$$\text{Out}[4] = \frac{1}{x} - \frac{x}{3} - \frac{x^3}{45} - \frac{2x^5}{945} + O[x]^6$$

A. Grozin, *Introduction to Mathematica® for Physicists*, Graduate Texts in Physics,
DOI 10.1007/978-3-319-00894-3_8, © Springer International Publishing Switzerland 2014

In[5] := FullForm[*s*]
Out[5]//FullForm =
SeriesData[*x*, 0, List[1, 0, Rational[−1, 3], 0, Rational[−1, 45], 0, Rational[−2, 945]], −1, 6, 1]

This is a series with half-integer degrees.

In[6] := *s* = Series[Sqrt[*x* ∗ (1 − *x*)], {*x*, 0, 5}]

$$\text{Out[6]} = \sqrt{x} - \frac{x^{3/2}}{2} - \frac{x^{5/2}}{8} - \frac{x^{7/2}}{16} - \frac{5x^{9/2}}{128} + O[x]^{11/2}$$

In[7] := FullForm[*s*]
Out[7]//FullForm =
SeriesData[*x*, 0, List[1, 0, Rational[−1, 2], 0, Rational[−1, 8], 0, Rational[−1, 16], 0, Rational[−5, 128]], 1, 11, 2]

This is an expansion at infinity.

In[8] := *s* = Series[Log[*x* + 1], {*x*, Infinity, 4}]

$$\text{Out[8]} = -\text{Log}\left[\frac{1}{x}\right] + \frac{1}{x} - \frac{1}{2x^2} + \frac{1}{3x^3} - \frac{1}{4x^4} + O\left[\frac{1}{x}\right]^5$$

In[9] := FullForm[*s*]
Out[9]//FullForm =
SeriesData[*x*, DirectedInfinity[1], List[Times[−1, Log[Power[*x*, −1]]], 1, Rational[−1, 2], Rational[1, 3], Rational[−1, 4]], 0, 5, 1]

Coefficients of a series in x may depend on x, but only weakly, weaker than any degree.

In[10] := *s* = Series[*x*^*x*, {*x*, 0, 3}]

$$\text{Out[10]} = 1 + \text{Log}[x]x + \frac{1}{2}\text{Log}[x]^2x^2 + \frac{1}{6}\text{Log}[x]^3x^3 + O[x]^4$$

In[11] := FullForm[*s*]
Out[11]//FullForm =
SeriesData[*x*, 0, List[1, Log[*x*], Times[Rational[1, 2], Power[Log[*x*], 2]], Times[Rational[1, 6], Power[Log[*x*], 3]]], 0, 4, 1]

In[12] := Clear[*s*]

Operations with Series

Let's take three series.

In[13] := sinx = Series[Sin[*x*], {*x*, 0, 7}]

$$\text{Out[13]} = x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + O[x]^8$$

In[14] := cosx = Series[Cos[*x*], {*x*, 0, 7}]

$$\text{Out[14]} = 1 - \frac{x^2}{2} + \frac{x^4}{24} - \frac{x^6}{720} + O[x]^8$$

In[15] := tanx = Series[Tan[*x*], {*x*, 0, 7}]

$$\text{Out[15]} = x + \frac{x^3}{3} + \frac{2x^5}{15} + \frac{17x^7}{315} + O[x]^8$$

Series can be added, multiplied, divided, etc.

In[16] := tanx * cosx

Out[16] $= x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + O[x]^8$

In[17] := sinx/cosx

Out[17] $= x + \frac{x^3}{3} + \frac{2x^5}{15} + \frac{17x^7}{315} + O[x]^8$

In[18] := sinx^2 + cosx^2

Out[18] $= 1 + O[x]^8$

If a series occurs as an argument of a function, the function is expanded automatically.

In[19] := Exp[sinx]

Out[19] $= 1 + x + \frac{x^2}{2} - \frac{x^4}{8} - \frac{x^5}{15} - \frac{x^6}{240} + \frac{x^7}{90} + O[x]^8$

In[20] := (1 − cosx)/*x*^2

Out[20] $= \frac{1}{2} - \frac{x^2}{24} + \frac{x^4}{720} + O[x]^6$

Here is an interesting method to expand a function in x.

In[21] := *X* = Series[*x*, {*x*, 0, 7}]

Out[21] $= x + O[x]^8$

In[22] := Sin[*X*]

Out[22] $= x - \frac{x^3}{6} + \frac{x^5}{120} - \frac{x^7}{5040} + O[x]^8$

In[23] := Clear[*X*]

Series can be differentiated and integrated.

In[24] := *D*[cosx, *x*]

Out[24] $= -x + \frac{x^3}{6} - \frac{x^5}{120} + O[x]^7$

In[25] := Integrate[tanx, *x*]

Out[25] $= \frac{x^2}{2} + \frac{x^4}{12} + \frac{x^6}{45} + \frac{17x^8}{2520} + O[x]^9$

A series (beginning from a small term) can be substituted for the expansion variable of another series. This is Sin[Tan[x]].

In[26] := st = sinx/.*x*−>tanx

Out[26] $= x + \frac{x^3}{6} - \frac{x^5}{40} - \frac{55x^7}{1008} + O[x]^8$

An alternative syntax.

In[27] := ComposeSeries[sinx, tanx]

Out[27] $= x + \frac{x^3}{6} - \frac{x^5}{40} - \frac{55x^7}{1008} + O[x]^8$

Let's subtract Tan[Sin[x]]; this expression is expanded automatically, i.e., series are contagious.

In[28] := st − Tan[Sin[*x*]]

Out[28] $= -\frac{x^7}{30} + O[x]^8$

In[29] := Clear[st]

Series inversion—solving the equation tanx $= y$ for x as a series in y.

In[30] := atany = InverseSeries[tanx, y]

$$\text{Out}[30] = y - \frac{y^3}{3} + \frac{y^5}{5} - \frac{y^7}{7} + O[y]^8$$

The result should be the arctangent.

In[31] := Series[ArcTan[y], {y, 0, 7}]

$$\text{Out}[31] = y - \frac{y^3}{3} + \frac{y^5}{5} - \frac{y^7}{7} + O[y]^8$$

In[32] := ComposeSeries[tanx, atany]

$\text{Out}[32] = y + O[y]^8$

It is not allowed to substitute a numerical value for the expansion variable into a series. The function Normal converts a series into a normal expression by dropping $+O[x]^n$. Here is a plot of sine and a few truncations of its series.

In[33] := Plot[Evaluate[Prepend[Table[Normal[Series[Sin[x], {x, 0, n}]], {n, 1, 5, 2}], Sin[x]]], {x, −Pi, Pi}]

Out[33] =



In[34] := Clear[sinx, cosx, tanx, atany]

You should work with series as long as possible, converting them into normal polynomials only at the very end. Then terms of too high orders of smallness are dropped automatically. At any moment you know exactly what is the order of the neglected term $O[x]^n$.

Arbitrary-Order Term

The function SeriesCoefficient can also be used in this way.

In[35] := SeriesCoefficient[Exp[x], {x, 0, 4}]

$$\text{Out}[35] = \frac{1}{24}$$

This is the 4th coefficient in the expansion of Exp[x] in x at 0. The number of the series term can be given symbolically.

In[36] := cn = SeriesCoefficient[Exp[x], {x,0,n}]
Out[36] = $\begin{cases} \frac{1}{n!} & n \geq 0 \\ 0 & \text{True} \end{cases}$
In[37] := Sum[cn ∗ x^n, {n,0,Infinity}]
Out[37] = e^x
In[38] := cn = SeriesCoefficient[Cos[x], {x,0,n}]
Out[38] = $\begin{cases} \frac{i^n(1+(-1)^n)}{2n!} & n \geq 0 \\ 0 & \text{True} \end{cases}$
In[39] := Sum[cn ∗ x^n, {n,0,Infinity}]
Out[39] = Cos[x]
In[40] := Clear[cn]

8.2 Differentiation

In[41] := f = x ∗ Sin[$x+y$]
Out[41] = xSin[$x+y$]
The derivative in x; in y; in x and y; the second derivative in x; the second derivative in x and the first one in y:
In[42] := {D[f,x],D[f,y],D[f,x,y],D[f,{x,2}],D[f,{x,2},y]}
Out[42] = {xCos[$x+y$] + Sin[$x+y$], xCos[$x+y$], Cos[$x+y$] − xSin[$x+y$],
2Cos[$x+y$] − xSin[$x+y$], −xCos[$x+y$] − 2Sin[$x+y$]}
In[43] := Clear[f]

Unknown Functions

Expressions with unknown functions can be differentiated.
In[44] := D[x ∗ f[x^2],x]
Out[44] = $f\left[x^2\right] + 2x^2 f'\left[x^2\right]$
Mathematica represents the first derivative of an unknown function f as the operator Derivative[1] applied to f.
In[45] := FullForm[%]
Out[45]//FullForm =
Plus[f[Power[x,2]], Times[2, Power[x,2], Derivative[1][f][Power[x,2]]]]
And this is the second derivative.
In[46] := Expand[D[x ∗ f[x^2], {x,2}]]
Out[46] = $6xf'\left[x^2\right] + 4x^3 f''\left[x^2\right]$
In[47] := FullForm[%]
Out[47]//FullForm =
Plus[Times[6, x, Derivative[1][f][Power[x,2]]],
Times[4, Power[x,3], Derivative[2][f][Power[x,2]]]]

Derivative[2,3] means the second derivative in the first argument and the third one in the second.
In[48] := $D[f[x,y],\{x,2\},\{y,3\}]$
Out[48] = $f^{(2,3)}[x,y]$
In[49] := FullForm[%]
Out[49]//FullForm =
Derivative[2,3][f][x,y]

Defining Derivatives

Let's tell *Mathematica* that the derivative of the function f is g.
In[50] := $f'[x\_] := g[x]$
In[51] := $D[x * f[x\verb|^|2],x]$
Out[51] = $f\left[x^2\right]+2x^2 g\left[x^2\right]$
The second derivative is not substituted automatically.
In[52] := Expand[$D[x * f[x\verb|^|2],\{x,2\}]$]
Out[52] = $6xg\left[x^2\right]+4x^3 f''\left[x^2\right]$
we can tell Mathematica that $\frac{\partial^3 f[x,y]}{\partial x \partial^2 y}$ is a function g.
In[53] := Derivative[1,2][f][$x\_,y\_$] := $g[x,y]$
In[54] := $D[x * f[x,y],\{x,2\},\{y,2\}]$
Out[54] = $2g[x,y]+xf^{(2,2)}[x,y]$

8.3 Integration

Indefinite Integrals

In[55] := Integrate[1/(x * (x^2 − 2)^2), x]
Out[55] = $-\dfrac{1}{4\left(-2+x^2\right)}+\dfrac{\text{Log}[x]}{4}-\dfrac{1}{8}\text{Log}\left[2-x^2\right]$
In[56] := Integrate[1/(Exp[x] + 1), x]
Out[56] = $x-\text{Log}\left[1+e^x\right]$
In[57] := Integrate[x/(Exp[x] + 1), x]
Out[57] = $\dfrac{x^2}{2}-x\text{Log}\left[1+e^x\right]-\text{PolyLog}\left[2,-e^x\right]$
In[58] := Integrate[Log[x], x]
Out[58] = $-x+x\text{Log}[x]$
In[59] := Integrate[1/Log[x], x]
Out[59] = LogIntegral[x]

In[60] := Integrate[Exp[x^2], x]

$\text{Out}[60] = \frac{1}{2}\sqrt{\pi}\,\text{Erfi}[x]$

In[61] := Integrate[x * Exp[x^2], x]

$\text{Out}[61] = \frac{e^{x^2}}{2}$

In[62] := Integrate[1/Sqrt[(1 − x^2) * (1 − k^2 * x^2)], x]

$\text{Out}[62] = \frac{\sqrt{1-x^2}\sqrt{1-k^2x^2}\,\text{EllipticF}\left[\text{ArcSin}[x],k^2\right]}{\sqrt{(-1+x^2)(-1+k^2x^2)}}$

In[63] := Simplify[Integrate[x/Sqrt[(1 − x^2) * (1 − k^2 * x^2)], x], x > 1]

$\text{Out}[63] = \frac{\text{Log}\left[k\left(k\sqrt{-1+x^2}+\sqrt{-1+k^2x^2}\right)\right]}{k}$

Definite Integrals

Here *Mathematica* produces a result with some assumptions about the parameter n.

In[64] := Integrate[x^n, {x, 0, 1}]

$\text{Out}[64] = \text{ConditionalExpression}\left[\frac{1}{1+n}, \text{Re}[n] > -1\right]$

Let's tell it that $n > -1$.

In[65] := Integrate[x^n, {x, 0, 1}, Assumptions−>{n > −1}]

$\text{Out}[65] = \frac{1}{1+n}$

In[66] := Integrate[Exp[a * Sin[x]], {x, 0, 2 * Pi}]

$\text{Out}[66] = 2\pi\,\text{BesselI}[0,a]$

In[67] := Integrate[Log[x]/(1 − x), {x, 0, 1}]

$\text{Out}[67] = -\frac{\pi^2}{6}$

The default value of the option Assumptions for Simplify, Integrate, etc. can be given in the variable $Assumptions.

In[68] := $Assumptions = {t > 0, t < 1, a > −1, b > −1};

In[69] := Integrate[x^a * (1 − x)^b * (1 − t * x)^c, {x, 0, 1}]

$\text{Out}[69] = -(\pi\,\text{Csc}[a\pi]\,\text{Gamma}[1+b]\,\text{Hypergeometric2F1Regularized}[1+a,-c, 2+a+b,t])/\text{Gamma}[-a]$

Now we can clear $Assumptions.

In[70] := $Assumptions = True;

Multiple integral

In[71] := Integrate[1/(1 + x * y), {x, 0, 1}, {y, 0, 1}]

$\text{Out}[71] = \frac{\pi^2}{12}$

8.4 Summation

Finite Sums

In[72] := Sum[n, {n, 0, k}]
Out[72] = $\frac{1}{2}k(1+k)$
In[73] := Sum[n^2, {n, 0, k}]
Out[73] = $\frac{1}{6}k(1+k)(1+2k)$
In[74] := Sum[x^n, {n, 0, k}]
Out[74] = $\frac{-1+x^{1+k}}{-1+x}$
In[75] := Sum[Binomial[k, n], {n, 0, k}]
Out[75] = 2^k
In[76] := Sum[(−1)^n ∗ Binomial[k, n], {n, 0, k}]
Out[76] = KroneckerDelta[k]
In[77] := Sum[Binomial[k, n]^2, {n, 0, k}]
Out[77] = Binomial[$2k, k$]

Series

In[78] := Sum[1/n^2, {n, 1, Infinity}]
Out[78] = $\frac{\pi^2}{6}$
In[79] := Sum[1/n^4, {n, 1, Infinity}]
Out[79] = $\frac{\pi^4}{90}$
In[80] := Sum[(−1)^n/n^2, {n, 1, Infinity}]
Out[80] = $-\frac{\pi^2}{12}$
In[81] := Sum[x^n/n!, {n, 0, Infinity}]
Out[81] = e^x

8.5 Differentiol Equations

A first-order differential equation.
In[82] := DSolve[D[x[t], t] + x[t] == 0, x[t], t]
Out[82] = $\{\{x[t] \to e^{-t}C[1]\}\}$
The solution contains an arbitrary constant $C[1]$. Let's add an initial condition:

In[83] := DSolve[{D[x[t],t] + x[t] == 0, x[0] == 1}, x[t], t]
Out[83] = $\{\{x[t] \to \mathrm{e}^{-t}\}\}$

A second-order differential equation.

In[84] := DSolve[D[x[t], {t, 2}] + x[t] == 0, x[t], t]
Out[84] = $\{\{x[t] \to C[1]\,\mathrm{Cos}[t] + C[2]\,\mathrm{Sin}[t]\}\}$

Initial conditions.

In[85] := DSolve[{D[x[t], {t, 2}] + x[t] == 0, x[0] == 0, x′[0] == 1}, x[t], t]
Out[85] = $\{\{x[t] \to \mathrm{Sin}[t]\}\}$

Boundary conditions.

In[86] := DSolve[{D[x[t], {t, 2}] + x[t] == 0, x[0] == 0, x[1] == 1}, x[t], t]
Out[86] = $\{\{x[t] \to \mathrm{Csc}[1]\,\mathrm{Sin}[t]\}\}$

A system of differential equations.

In[87] := DSolve[{D[x[t], t] == p[t], D[p[t], t] == −x[t]}, {x[t], p[t]}, t]
Out[87] = $\{\{p[t] \to C[1]\,\mathrm{Cos}[t] - C[2]\,\mathrm{Sin}[t], x[t] \to C[2]\,\mathrm{Cos}[t] + C[1]\,\mathrm{Sin}[t]\}\}$